title: GLSL Shadows with Perlin Noise
body: <p>So I've recently gotten into using WebGL and more specifically writing GLSL Shaders and I have run into a snag while writing the fragment shader for my "water" shader which is derived from <a href="http://www.lighthouse3d.com/tutorials/glsl-tutorial/toon-shader-version-ii/" rel="nofollow noreferrer">this tutorial.</a></p>
  
  <p>What I'm trying to achieve is a stepped shading (Toon shading, cell shading...) effect on  waves generated by my vertex shader but the fragment shader seems to treat the waves as though they are still a flat plane and the entire mesh is drawn as one solid color. </p>
  
  <p>What am I missing here? The sphere works perfectly but flat surfaces are all shaded uniformly. I have the same problem if I use a cube. Each face on the cube is shaded independently but the entire face is given a solid color.</p>
  
  <h2>The Scene</h2>
  
  <p>This is how I have my test scene set up. I have two meshes using the same material - a sphere and a plane and a light source.</p>
  
  <p><img src="https://i.stack.imgur.com/hYogb.jpg" alt="The setup"></p>
  
  <h2>The Problem</h2>
  
  <p>As you can see the shader is working as expected on the sphere. 
  I enabled wireframe for this shot to show that the vertex shader (perlin noise) is working beautifully on the plane.</p>
  
  <p><img src="https://i.stack.imgur.com/lr7li.jpg" alt="wireframe enabled to illustrate noise"></p>
  
  <p>But when I turn the wireframe off you can see that the fragment shader seems to be receiving the same level of light uniformly across the entire plane creating this...</p>
  
  <p><img src="https://i.stack.imgur.com/ueMQK.jpg" alt="enter image description here"></p>
  
  <p>Rotating the plane to face the light source will change the color of the material but again the color is applied uniformly over the entire surface of the plane.
  <img src="https://i.stack.imgur.com/IxyJB.jpg" alt="enter image description here"></p>
  
  <h1>The Fragment Shader</h1>
  
  <p>In all it's script kid glory lol. </p>
  
  <pre><code>uniform vec3 uMaterialColor;
  uniform vec3 uDirLightPos;
  uniform vec3 uDirLightColor;
  uniform float uKd;
  uniform float uBorder;
  varying vec3 vNormal;
  varying vec3 vViewPosition;
  
  void main() {
  
      vec4 color;
  
      // compute direction to light
      vec4 lDirection = viewMatrix * vec4( uDirLightPos, 0.0 );
      vec3 lVector = normalize( lDirection.xyz );
  
      //  N * L. Normal must be normalized, since it's interpolated.
      vec3 normal = normalize( vNormal );
  
      // check the diffuse dot product against uBorder and adjust
      // this diffuse value accordingly.
      float diffuse = max( dot( normal, lVector ), 0.0);
  
      if (diffuse &gt; 0.95)
          color = vec4(1.0,0.0,0.0,1.0);
      else if (diffuse &gt; 0.85)
          color = vec4(0.9,0.0,0.0,1.0);
      else if (diffuse &gt; 0.75)
          color = vec4(0.8,0.0,0.0,1.0);
      else if (diffuse &gt; 0.65)
          color = vec4(0.7,0.0,0.0,1.0);
      else if (diffuse &gt; 0.55)
          color = vec4(0.6,0.0,0.0,1.0);
      else if (diffuse &gt; 0.45)
          color = vec4(0.5,0.0,0.0,1.0);
      else if (diffuse &gt; 0.35)
          color = vec4(0.4,0.0,0.0,1.0);
      else if (diffuse &gt; 0.25)
          color = vec4(0.3,0.0,0.0,1.0);
      else if (diffuse &gt; 0.15)
          color = vec4(0.2,0.0,0.0,1.0);
      else if (diffuse &gt; 0.05)
          color = vec4(0.1,0.0,0.0,1.0);
      else
          color = vec4(0.05,0.0,0.0,1.0);
  
      gl_FragColor = color;
  </code></pre>
  
  <h1>The Vertex Shader</h1>
  
  <pre><code>    vec3 mod289(vec3 x)
  {
      return x - floor(x * (1.0 / 289.0)) * 289.0;
  }
  
  vec4 mod289(vec4 x)
  {
      return x - floor(x * (1.0 / 289.0)) * 289.0;
  }
  
  vec4 permute(vec4 x)
  {
      return mod289(((x*34.0)+1.0)*x);
  }
  
  vec4 taylorInvSqrt(vec4 r)
  {
      return 1.79284291400159 - 0.85373472095314 * r;
  }
  
  vec3 fade(vec3 t) {
      return t*t*t*(t*(t*6.0-15.0)+10.0);
  }
  
  // Classic Perlin noise
  float cnoise(vec3 P)
  {
      vec3 Pi0 = floor(P); // Integer part for indexing
      vec3 Pi1 = Pi0 + vec3(1.0); // Integer part + 1
      Pi0 = mod289(Pi0);
      Pi1 = mod289(Pi1);
      vec3 Pf0 = fract(P); // Fractional part for interpolation
      vec3 Pf1 = Pf0 - vec3(1.0); // Fractional part - 1.0
      vec4 ix = vec4(Pi0.x, Pi1.x, Pi0.x, Pi1.x);
      vec4 iy = vec4(Pi0.yy, Pi1.yy);
      vec4 iz0 = Pi0.zzzz;
      vec4 iz1 = Pi1.zzzz;
  
      vec4 ixy = permute(permute(ix) + iy);
      vec4 ixy0 = permute(ixy + iz0);
      vec4 ixy1 = permute(ixy + iz1);
  
      vec4 gx0 = ixy0 * (1.0 / 7.0);
      vec4 gy0 = fract(floor(gx0) * (1.0 / 7.0)) - 0.5;
      gx0 = fract(gx0);
      vec4 gz0 = vec4(0.5) - abs(gx0) - abs(gy0);
      vec4 sz0 = step(gz0, vec4(0.0));
      gx0 -= sz0 * (step(0.0, gx0) - 0.5);
      gy0 -= sz0 * (step(0.0, gy0) - 0.5);
  
      vec4 gx1 = ixy1 * (1.0 / 7.0);
      vec4 gy1 = fract(floor(gx1) * (1.0 / 7.0)) - 0.5;
      gx1 = fract(gx1);
      vec4 gz1 = vec4(0.5) - abs(gx1) - abs(gy1);
      vec4 sz1 = step(gz1, vec4(0.0));
      gx1 -= sz1 * (step(0.0, gx1) - 0.5);
      gy1 -= sz1 * (step(0.0, gy1) - 0.5);
  
      vec3 g000 = vec3(gx0.x,gy0.x,gz0.x);
      vec3 g100 = vec3(gx0.y,gy0.y,gz0.y);
      vec3 g010 = vec3(gx0.z,gy0.z,gz0.z);
      vec3 g110 = vec3(gx0.w,gy0.w,gz0.w);
      vec3 g001 = vec3(gx1.x,gy1.x,gz1.x);
      vec3 g101 = vec3(gx1.y,gy1.y,gz1.y);
      vec3 g011 = vec3(gx1.z,gy1.z,gz1.z);
      vec3 g111 = vec3(gx1.w,gy1.w,gz1.w);
  
      vec4 norm0 = taylorInvSqrt(vec4(dot(g000, g000), dot(g010, g010), dot(g100, g100), dot(g110, g110)));
      g000 *= norm0.x;
      g010 *= norm0.y;
      g100 *= norm0.z;
      g110 *= norm0.w;
      vec4 norm1 = taylorInvSqrt(vec4(dot(g001, g001), dot(g011, g011), dot(g101, g101), dot(g111, g111)));
      g001 *= norm1.x;
      g011 *= norm1.y;
      g101 *= norm1.z;
      g111 *= norm1.w;
  
      float n000 = dot(g000, Pf0);
      float n100 = dot(g100, vec3(Pf1.x, Pf0.yz));
      float n010 = dot(g010, vec3(Pf0.x, Pf1.y, Pf0.z));
      float n110 = dot(g110, vec3(Pf1.xy, Pf0.z));
      float n001 = dot(g001, vec3(Pf0.xy, Pf1.z));
      float n101 = dot(g101, vec3(Pf1.x, Pf0.y, Pf1.z));
      float n011 = dot(g011, vec3(Pf0.x, Pf1.yz));
      float n111 = dot(g111, Pf1);
  
      vec3 fade_xyz = fade(Pf0);
      vec4 n_z = mix(vec4(n000, n100, n010, n110), vec4(n001, n101, n011, n111), fade_xyz.z);
      vec2 n_yz = mix(n_z.xy, n_z.zw, fade_xyz.y);
      float n_xyz = mix(n_yz.x, n_yz.y, fade_xyz.x); 
      return 2.2 * n_xyz;
  }
  
  // Classic Perlin noise, periodic variant
  float pnoise(vec3 P, vec3 rep)
  {
      vec3 Pi0 = mod(floor(P), rep); // Integer part, modulo period
      vec3 Pi1 = mod(Pi0 + vec3(1.0), rep); // Integer part + 1, mod period
      Pi0 = mod289(Pi0);
      Pi1 = mod289(Pi1);
      vec3 Pf0 = fract(P); // Fractional part for interpolation
      vec3 Pf1 = Pf0 - vec3(1.0); // Fractional part - 1.0
      vec4 ix = vec4(Pi0.x, Pi1.x, Pi0.x, Pi1.x);
      vec4 iy = vec4(Pi0.yy, Pi1.yy);
      vec4 iz0 = Pi0.zzzz;
      vec4 iz1 = Pi1.zzzz;
  
      vec4 ixy = permute(permute(ix) + iy);
      vec4 ixy0 = permute(ixy + iz0);
      vec4 ixy1 = permute(ixy + iz1);
  
      vec4 gx0 = ixy0 * (1.0 / 7.0);
      vec4 gy0 = fract(floor(gx0) * (1.0 / 7.0)) - 0.5;
      gx0 = fract(gx0);
      vec4 gz0 = vec4(0.5) - abs(gx0) - abs(gy0);
      vec4 sz0 = step(gz0, vec4(0.0));
      gx0 -= sz0 * (step(0.0, gx0) - 0.5);
      gy0 -= sz0 * (step(0.0, gy0) - 0.5);
  
      vec4 gx1 = ixy1 * (1.0 / 7.0);
      vec4 gy1 = fract(floor(gx1) * (1.0 / 7.0)) - 0.5;
      gx1 = fract(gx1);
      vec4 gz1 = vec4(0.5) - abs(gx1) - abs(gy1);
      vec4 sz1 = step(gz1, vec4(0.0));
      gx1 -= sz1 * (step(0.0, gx1) - 0.5);
      gy1 -= sz1 * (step(0.0, gy1) - 0.5);
  
      vec3 g000 = vec3(gx0.x,gy0.x,gz0.x);
      vec3 g100 = vec3(gx0.y,gy0.y,gz0.y);
      vec3 g010 = vec3(gx0.z,gy0.z,gz0.z);
      vec3 g110 = vec3(gx0.w,gy0.w,gz0.w);
      vec3 g001 = vec3(gx1.x,gy1.x,gz1.x);
      vec3 g101 = vec3(gx1.y,gy1.y,gz1.y);
      vec3 g011 = vec3(gx1.z,gy1.z,gz1.z);
      vec3 g111 = vec3(gx1.w,gy1.w,gz1.w);
  
      vec4 norm0 = taylorInvSqrt(vec4(dot(g000, g000), dot(g010, g010), dot(g100, g100), dot(g110, g110)));
      g000 *= norm0.x;
      g010 *= norm0.y;
      g100 *= norm0.z;
      g110 *= norm0.w;
      vec4 norm1 = taylorInvSqrt(vec4(dot(g001, g001), dot(g011, g011), dot(g101, g101), dot(g111, g111)));
      g001 *= norm1.x;
      g011 *= norm1.y;
      g101 *= norm1.z;
      g111 *= norm1.w;
  
      float n000 = dot(g000, Pf0);
      float n100 = dot(g100, vec3(Pf1.x, Pf0.yz));
      float n010 = dot(g010, vec3(Pf0.x, Pf1.y, Pf0.z));
      float n110 = dot(g110, vec3(Pf1.xy, Pf0.z));
      float n001 = dot(g001, vec3(Pf0.xy, Pf1.z));
      float n101 = dot(g101, vec3(Pf1.x, Pf0.y, Pf1.z));
      float n011 = dot(g011, vec3(Pf0.x, Pf1.yz));
      float n111 = dot(g111, Pf1);
  
      vec3 fade_xyz = fade(Pf0);
      vec4 n_z = mix(vec4(n000, n100, n010, n110), vec4(n001, n101, n011, n111), fade_xyz.z);
      vec2 n_yz = mix(n_z.xy, n_z.zw, fade_xyz.y);
      float n_xyz = mix(n_yz.x, n_yz.y, fade_xyz.x); 
      return 2.2 * n_xyz;
  }
  
  varying vec2 vUv;
  varying float noise;
  uniform float time;
  
  // for the cell shader
  varying vec3 vNormal;
  varying vec3 vViewPosition;
  
  float turbulence( vec3 p ) {
      float w = 100.0;
      float t = -.5;
      for (float f = 1.0 ; f &lt;= 10.0 ; f++ ){
          float power = pow( 2.0, f );
          t += abs( pnoise( vec3( power * p ), vec3( 10.0, 10.0, 10.0 ) ) / power );
      }
      return t;
  }
  
  varying vec3 vertexWorldPos;
  
  void main() {
  
      vUv = uv;
  
      // add time to the noise parameters so it's animated
      noise = 10.0 *  -.10 * turbulence( .5 * normal + time );
      float b = 25.0 * pnoise( 0.05 * position + vec3( 2.0 * time ), vec3( 100.0 ) );
      float displacement = - 10. - noise + b;
  
      vec3 newPosition = position + normal * displacement;
      gl_Position = projectionMatrix * modelViewMatrix * vec4( newPosition, 1.0 );
  
      // for the cell shader effect
      vNormal = normalize( normalMatrix * normal );
      vec4 mvPosition = modelViewMatrix * vec4( position, 1.0 );
      vViewPosition = -mvPosition.xyz;
  }
  </code></pre>
  
  <h2>Worth Mention</h2>
  
  <p>I am using the Three.js library
  My light source is an instance of THREE.SpotLight</p>